
ZPL Commands

This section contains the complete alphabetical listing of ZPL II commands.

Description  This heading provides an explanation of how the command is used, what it is 
capable of, and any defining characteristics it has.

Format  Format explains how the command is syntactically arranged and what parameters it 
contains. For example, the ^B8 command prints a EAN-8 bar code. The format of the ^B8 
command is: ^B8o,h,f,g. It is arranged with the caret symbol (^), the command code (B8), 
and the parameters and are replaced with supported values. 

Parameters  If a command has values that can be defined to make its function more specific, 
these are outlined as parameters. Parameters typically have Accepted Values and Default 
Values. 

Still using the ^B8 example, the h parameter is defined as:

h = bar code height (in dots)
Accepted Values: 1 to 32000

Default Value: value set by ^BY

If the command has no parameters – for example ~JA (Cancel All) – the parameter heading is 
removed, indicating that the format of the command (~JA) is acceptable ZPL II code.

Comment  This section is reserved for notes that are of value to a programmer, warnings of 
potential command interactions, or command-specific information that should be taken into 
consideration. 

Basic ZPL Exercises and Examples
The purpose of these exercises is to introduce basic ZPL commands to novice ZPL users.

Make sure this checklist is complete:
Load the printer with labels that are big enough to give you ample space to work with.

Print a configuration label (CANCEL test). 

Look at the configuration label and make sure that the LEFT POSITION is set to 000 and 
LABEL TOP is set to 000. 

Determine the printer’s resolution. It is listed on the configuration label. 8/MM = 200 
dpi, 12/MM = 300 dpi and 24/MM = 600 dpi. 

Tips

These are some tips when using ZPL:

• Use the DOS text editor to write ZPL files.

• Save the file as a .txt file and copy it to the printer from DOS command line.

Example • When the command is best clarified in context, an example of the 
ZPL II code is provided. Text indicating exact code entered is printed in an easily 
recognizable Courier font. An example of code using the ^B8 command looks like this:

^XA

^FO50,50

^B8N,100,Y,N

^FD1234567^FS

^XZ
Notice that the ^B8 parameter letters have been replaced with real values that apply to the 
command. In this example N,100,Y,N have been entered.

Example • An example comment is: This command works only when the printer is idle, or 
This command is ignored if a value exceeds the parameter limits.
Comments are also included next to parameters if they apply directly to a particular setting.

Before you begin

Some things that are important to understand before you begin are:

• 200 dpi means the resolution of the printhead is 200 dots per inch. If you program the 
printer to draw a line 100 dots long that equals a half inch. 100 dots on a 
300 dpi printer prints a line 1/3 inch long.

• The home position that all your coordinates are referencing is at the left-hand trailing edge 
of the label as the label comes out of the printer. (There are some exceptions to this.)

Exercises

The exercises start simple and gradually progress to give you an opportunity to try a variety of 
commonly used ZPL commands. Not all commands are covered, but this should be a good 
core of commands to learn. Some commands may not be supported due to the firmware 
version in your printer.

Exercise 1 • This exercise shows you how to specify a location for an entered name.

1. Print your name on the label.

2. Start by printing just your name on the label. Use this format as a model: 

3. Send this format to the printer:

4. When the label prints correctly, alter the first number after the ^FOx. See how that change 
affects the print position. Alter the second number after the ^FO50,x and see how that the 
print position.

Important • Your name goes where you see xxxxxxxxxxx in the second line of code.

1 Every format starts with the ^XA command
2 ^FO (field origin) command
3 ^FS (field separator) command
4 Every format ends with the ^XZ command
5 ^FD (field data) command

^XA

^FO50,50^ADN,36,20^FDxxxxxxxxxxx^FS
^XZ

1

5

3
4

2

Font instruction

^ADN

1. Alter the numbers after the ^ADN,x,x command.

• 18,10 is the smallest size you can make the D font.

• The first number is the height of the font in dots. The second number is the width in 
dots.

• You can use direct multiples up to ten times that size as a maximum. 

• 25,18 would not be a valid size. The printer rounds to the next recognizable size.

2. Check the font matrices tables for other fonts to try. See Fonts and Bar Codes 
on page 1183.

3. Try the zero scalable font ^A0N,x,x. 
This font is scalable, and you can choose any height and width.

Rotation commands

1. Change ^ADN to ^ADR, and then ^ADI, and then ^ADB.
See how the print position changes.

2. Add more fields.

3. Add two more fields to print directly under your name using the ^ADN,36,20 font and size:
Your street address
Your city, state, zip

4. You must add two more lines of code that start off with:
^XA

^FO50,50^ADN,36,20^FDxxxxxxxxxxx^FS

^FO (fill in the rest)

^FO (fill in the rest)

^XZ

Make sure all these fields print in the same font and size and left side of fields has same 
vertical alignment.

Your name
1200 W Main Street
Anytown, Il 60061

Example • 180,100 is the largest you can make the D font.

Special Effects commands

The Graphic Box or ^GB command or is used in some of the speical effects commands. 

Reverse Printing a Field

1. Write the following format and send to the printer:
^XA

^PR1

^FO100,100

^GB70,70,70,,3^FS

^FO200,100

^GB70,70,70,,3^FS

^FO300,100

^GB70,70,70,,3^FS

^FO400,100

^GB70,70,70,,3^FS

^FO107,110^CF0,70,93

^FR^FDREVERSE^FS

^XZ

2. To see the effects, remove: 
^FR^FDREVERSE^FS

3. To see the effects, try removing one of the ^GB lines of code. 

Label Reverse Print

1. Write the following format and send to the printer:
^XA^LRY

^FO100,50

^GB195,203,195^FS

^FO180,110^CFG

^FDLABEL^FS

^FO130,170

^FDREVERSE^FS

^XZ

2. To see the effects, remove: 
^GB195,203,195^FS

Mirror Image of Label

1. Write the following format and send to the printer:
^XA^PMY

^FO100,100

^CFG

^FDMIRROR^FS

^FO100,160

^FDIMAGE^FS

^XZ

2. To see the effects, in the first line of code change ^PMY to ^PMN.

Print Orientation

1. Write the following format and send to the printer:
^XA^CFD

^POI

^LH330,10

^FO50,50

^FDZEBRA TECHNOLOGIES^FS

^FO50,75

^FDVernon Hills, IL^FS

^XZ

2. To see the effects, in the second line of code change ^POI to ^PON.

Exercise 2 • Boxes and lines

1. Use the address format from Exercise 1. 

2. Add this new line to your existing format:
^FO50,200^GB200,200,2^FS

This prints a box one wide by one inch long and the thickness of the line is 2 dots.

3. Reposition and resize the square so that it goes around the name and address uniformly.

4. Print a line by adding:
^FO50,300^GB400,0,4,^FS

This prints a horizontal line two inches wide by 4 dots thick.

5. Print a vertical line using this code:
^F0100,50^GBO,400,4^FS

Exercise 3 • Bar codes — ^B3 code 39 bar code

1. Write the following format and send to the printer:
^XA

^FO50,50^B3N,N,100,Y,N^FD123456^FS

^XZ

2. Try changing each of the parameters in the ^B3 string so you can see the effects.
^B3o,e,h,f,g

^BY

3. Insert the ^BY command just before the ^B3 to see how the narrow bar width can be altered.
^FO50,50^BY2^B3..etc ^BYx, acceptable values for x are 1 through 10

4. Alter the ratio of the narrow to wide bar.
^FO50,50^BY2,3^B3..etc ^BY2,x acceptable values for x are 2.1 through 3 in .1
increments

5. Print out a ^B3 bar code with the interpretation line on top of the bar code and the bar code 
rotated 90 degrees.

6. Add a ^PQ just before the ^XZ to print several labels.
^PQ4

^XZ
^PR Print rate (in inches per second)

7. Add a ^PR command after the ^XA at the beginning of the format to change the print rate 
(print speed).
^XA

^PR4 then try ^PR6 ^PRx acceptable values for x are 2 through 12 (check
printer specs)

See how the print speed affects the print quality of the bar code. You may need to increase 
the printer darkness setting at higher print speeds.

Important • For valid parameter choices, see ^B3 on page 68.

Exercise 4 • ^SN — Serial Number command

1. Send this format to the printer:
^XA

^FO100,100^ADN,36,20^SN001,1,Y^FS

^PQ3

^XZ

To vary the ^SNv,n,z to exercise increment/decrement and leading zeros functions, 
consult this guide.
If your serial number contains alpha and numeric characters, you can increment or 
decrement a specific segment of the data even if it is in the middle, as this sample 
sequence shows:
ABCD1000EFGH, ABCD1001EFGH, ABCD1002EFGH

2. Send this file to the printer and to see how it increments the serial number. The ^SF 
command can also work with alpha characters.
^XA

^FO100,100^ADN,36,20^FDABCD1000EFGH^SF%%%%dddd%%%%,10000^FS
^PQ15

^XZ

Notice how the field data character position aligns with the ^SF data string:

And on through…

The last label prints ABCD1014EFGH. 

The % is placed in positions that you do not want to increment or decrement, 
d = decimal, 10000 = increment value.

For more details on ^SF, see ^SF on page 341.

^ F D A B C D 1 0 0 0 E F G H
^ S F % % % % d d d d % % % %

1 0 0 0 0
2 0 0 0 0
3 0 0 0 0

1 0 1 4 0 0 0 0

Exercise 5 • Saving a template to memory. ^IS and image save and image load.

1. Send this format to the printer:
^XA

^FO20,30^GB750,1100,4^FS

^FO20,30^GB750,200,4^FS

^FO20,30^GB750,400,4^FS

^FO20,30^GB750,700,4^FS

^FO20,226^GB325,204,4^FS

^FO30,40^ADN,36,20^FDShip to:^FS

^FO30,260^ADN,18,10^FDPart number #^FS

^FO360,260^ADN,18,10^FDDescription:^FS

^FO30,750^ADN,36,20^FDFrom:^FS

^ISR:SAMPLE.GRF^FS

^XZ

2. Send this format:
^XA

^ILR:SAMPLE.GRF^FS

^FO150,125^ADN,36,20^FDAcme Printing^FS

^FO60,330^ADN,36,20^FD14042^FS

^FO400,330^ADN,36,20^FDScrew^FS

^FO70,480^BY4^B3N,,200^FD12345678^FS

^FO150,800^ADN,36,20^FDMacks Fabricating^FS

^XZ

In this way the template only needs to be sent one time to the printer’s memory. Subsequent 
formats can be sent recalling the template and merging variable data into the template. In this 
exercise, the file was saved in the printers R: memory, which is volatile.

Note • This exercise helps you troubleshoot your code against the errors you see on your 
labels.

Exercise 6 • ^DF and ^XF — Download format and recall format

Similar concept to ^IS and ^IL command. ^IS and ^IL processes faster in the printer than ^DF 
and ^XF. 

This is how the ^DF and ^XF format structure produces a label similar to the ^IS/^IL sample 
you just tried.

Figure 2 • Download and Recall Format

1 Download format code
2 Recall format call that generates the generated label in Figure 2.

^XA
^DFR:SAMPLE.GRF^FS
^FO20,30^GB750,1100,4^FS
^FO20,30^GB750,200,4^FS
^FO20,30^GB750,400,4^FS
^FO20,30^GB750,700,4^FS
^FO20,226^GB325,204,4^FS
^FO30,40^ADN,36,20^FDShip to:^FS
^FO30,260^ADN,18,10^FDPart number #^FS
^FO360,260^ADN,18,10^FDDescription:^FS
^FO30,750^ADN,36,20^FDFrom:^FS
^FO150,125^ADN,36,20^FN1^FS (ship to)
^FO60,330^ADN,36,20^FN2^FS(part num)
^FO400,330^ADN,36,20^FN3^FS(description)
^FO70,480^BY4^B3N,,200^FN4^FS(barcode)
^FO150,800^ADN,36,20^FN5^FS (from)
^XZ

^XA
^XFR:SAMPLE.GRF
^FN1^FDAcme Printing^FS
^FN2^FD14042^FS
^FN3^FDScrew^FS
^FN4^FD12345678^FS
^FN5^FDMacks Fabricating^FS
^XZ

ZPL II CODE GENERATED LABEL

1

2

Exercise 7 • Asian and Unicode Encodings

In each of the following examples, the format is saved in the corresponding encoding before 
being sent down to the printer and the ZPL script was made in Microsoft Notepad, a basic text 
editor. The characters were inserted from the character map in Windows or typed from the 
keyboard. 

This exercise works in printers with firmware version V60.14, V50.14, or later.

Example • This is an example of using an Asian encoding, such as UHANGUL, with ASCII 
text. Using the CI26 command tells the printer to recognize any byte less than 7F as ASCII 
text and every byte above as the first byte of UHANGUAL encoding:

Example • This is an example of using the Unicode encoding, UTF-8:

ZPL II CODE GENERATED LABEL

ZPL II CODE

GENERATED LABEL


